


Informatique Systèmes d'Information Bases de Données



boufares@iutv.univ-paris13.fr
<http://www.lipn.univ-paris13.fr/~boufares>
Bureau Q304 ; tél: +33 (0)1 49 40 28 50
Université Paris 13, Sorbonne Paris Cité
IUT Villetaneuse
99, Avenue J.B. Clément
F-93430 Villetaneuse FRANCE

boufares@lipn.univ-paris13.fr
<http://www.lipn.univ-paris13.fr/~boufares>
Bureau A109 ; tél: +33 (0)1 49 40 40 71
Laboratoire d'Informatique de Paris Nord
LIPN – UMR 7030 du CNRS
Université Paris 13, Sorbonne Paris Cité
Institut Galilée
99, Avenue J.B. Clément
F-93430 Villetaneuse France
www.univ-paris13.fr

Polycopié N° 2



**Aujourd'hui,
j'arrête de fumer !**

Le TABAC t'ABAT !



Le TABAC t'ABAT !

**FUMER NUIT GRAVEMENT
À VOTRE SANTÉ ET
À CELLE DE VOTRE ENTOURAGE !**



M T Dents !

SMILE and the World SMILES with you !



Le SPORT est Générateur de CONFIANCE !



Informatique ; Bases de Données et Systèmes d'Information

M. F. BOUFARES
UNIVERSITE PARIS 13, Sorbonne Paris Cité
IUT Villetaneuse

SI() ; SI(SI()) ; SI(SI(SI())) ;

NB.SI() ; SOMME.SI() ; MOYENNE.SI()

NB.SI.ENS() ; SOMME.SI.ENS() ; MOYENNE.SI.ENS()

**{NB(SI())} ; {SOMME(SI())} ; {MOYENNE(SI())} ;
{MIN(SI())} ; {MAX(SI())} ;
{VARP(SI())} ; ...**



1. La fonction logique d'EXCEL **SI()**

La **fonction logique SI**, dans EXCEL, commence par le signe égal (=). Elle **ne doit pas comporter d'espace** (sauf si l'on utilise des guillemets, pour utiliser une donnée du type texte). Elle peut être de la forme suivante :

SI **condition** ALORS **ActionOui** SINON **ActionNon** FINSI

=SI(**condition**;**ActionOui**;**ActionNon**)

L'**actionOui** est déclenchée si la **condition** est **vérifiée**, dans **le cas contraire c'est l'actionNon** qui est exécutée (celle-ci peut être vide).

L'ActionOui ou l'ActionNon ou les deux à la fois peuvent comporter des fonctions logiques SI, il s'agit alors d'une imbrication.

Exemples : =SI(N₁>15;"Bonne note";"Peut mieux faire")
 Avec N₁ = 17 le résultat affiché sera → Bonne note



Exemples :

=SI(**$N_1 > 15$** ;"Bonne note";"Peut mieux faire")

Avec $N_1 = 17$ le résultat affiché sera → Bonne note

=SI(**$N_1 \leq 15$** ;"Peut mieux faire";"Bonne note")

Avec $N_1 = 17$ le résultat affiché sera → Bonne note

=SI(**NON(ESTTEXTE(A_7))**;"Donnée numérique";"Donnée texte")

Avec A_7 qui contient le mot Paris le résultat affiché sera → Donnée texte

=SI(**ESTNUM(A_7)**;"Donnée numérique";"Donnée texte")


Avec A_7 qui contient le mot Paris le résultat affiché sera → Donnée texte



Comment écrire la condition (le critère de sélection) ?

La condition (le critère) doit être une proposition logique (bouléenne) évaluée à VRAI ou à FAUX.

Elle peut être composée de trois parties :



Cellule veut dire « une référence à une cellule ».

Exemples : B7 ; \$A1 etc...

Valeur veut dire une valeur constante ou une valeur calculée qui fait appel à une fonction XL.

Exemples : 7 ; VRAI ; "Bonne note" ; 2% ; GAUCHE(F7;2) ; NB(A1:A7) etc...



La fonction SI() :

ÉCRITURE ALGORITHMIQUE	ÉCRITURE EXCEL
SI Condition ALORS ActionOui SINON ActionNon FINSI	=SI(Condition;ActionOui;ActionNon)
SI Condition ALORS ActionOui FINSI <i>Remarque : La deuxième action n'est pas spécifiée !</i>	=SI(Condition;ActionOui;"") Les guillemets doivent exister en l'absence de l'actionNon.

Remarque : Etant donné la formule : =SI(Condition;ActionOui)

Cette formule ne précise pas ce qu'il y à faire quand la condition n'est pas vérifiée (ceci est traduit par l'absence du 2^{ème} ;). Cette formule sera donc interprétée par la machine comme suit:

- l'action du oui est exécutée lorsque la condition est vérifiée ;
- le résultat affiché par la machine est **FAUX (ou égal à 0 !!!)**, si la condition n'est pas vérifiée.

Exemple : =SI(N₁>15;"Bonne note") ; Avec N₁ = 10 ; le résultat affiché sera → FAUX !!!

Exemple : =SI(N₁>15;"Bonne note" ;) Avec N₁ = 10 ; le résultat affiché sera → 0 !!!



Les tests peuvent être **imbriqués** (les uns dans les autres) comme l'indique le tableau ci-dessous.

Imbrication des SI :

ÉCRITURE ALGORITHMIQUE	ÉCRITURE EXCEL
SI Condition1 ALORS	=SI(Condition1 ; SI(Condition2;ActionOui2;ActionNon2)); ActionNon1)
SI Condition2	
ALORS ActionOui2	
SINON ActionNon2	
FINSI	
SINON ActionNon1 FINSI	=SI(Condition1 ;ActionOui1; SI(Condition2;ActionOui2;ActionNon2))
SI Condition1 ALORS ActionOui1 SINON	
SI Condition2	
ALORS ActionOui2	
SINON ActionNon2	
FINSI	
FINSI	



Les tests peuvent être **imbriqués** (les uns dans les autres) comme l'indique le tableau ci-dessous.
 Imbrication des SI :

ÉCRITURE ALGORITHMIQUE	ÉCRITURE EXCEL
SI Condition1	=SI(Condition1 ; SI(Condition2 ;ActionOui2;ActionNon2); SI(Condition3 ;ActionOui3;ActionNon3))
ALORS	
SI Condition2	
ALORS ActionOui2	
SINON ActionNon2	
FINSI	
SINON	
SI Condition3	
ALORS ActionOui3	
SINON ActionNon3	
FINSI	
FINSI	



Les tests peuvent être **imbriqués** (les uns dans les autres) comme l'indique le tableau ci-dessous.
 Imbrication des SI :

ÉCRITURE ALGORITHMIQUE	ÉCRITURE EXCEL
<pre> SI Condition1 ALORS ActionOui1 SINON SI Condition2 ALORS ActionOui2 SINON SI Condition3 ALORS ActionOui3 ... FINSI FINSI FINSI </pre>	<pre> =SI(Condition1 ; ActionOui1 ; SI(Condition2 ; ActionOui2 ; SI(Condition3 ; ActionOui3 ; ...))) </pre>



Conditions multicritères : Dans le cas où la condition porte sur plusieurs critères (on parle alors d'expression logique complexe) deux opérateurs logiques peuvent être utilisés : le ET et le OU comme suit :

Dans le cas où **tous les critères doivent être vérifiés** pour que l'ActionOui soit exécutée (l'ActionNon n'est exécutée que si au moins un critère n'est pas vérifié) :

La condition dit que le **Critère1 ET le Critère2 ET ...** doivent **tous** être vérifiés

ÉCRITURE ALGORITHMIQUE	ÉCRITURE EXCEL
(Critère1 ET Critère2 ET ...)	ET (Critère1 ; Critère2 ; ...)

Dans le cas où **au moins un critère doit être vérifié** pour que l'ActionOui soit exécutée (l'ActionNon n'est exécutée que si aucun critère n'est vérifié) :

La condition dit qu'**au moins le Critère1 OU le Critère2 OU ...** est vérifié

ÉCRITURE ALGORITHMIQUE	ÉCRITURE EXCEL
(Critère1 OU Critère2 OU...)	OU (Critère1 ; Critère2 ; ...)



	A	B	C	D	E
1	Le ET logique :				
2	Deux ou plusieurs critères :				
3	ET(Critère1;Critère2)				
4	ET(Critère1;Critère2; ... ; CritèreN)				
5					
6	Données	Formules	Résultats		
7	17	=A\$7>A\$8	VRAI		
8	9	=A\$8>A\$9	FAUX		
9	58	=A\$9=(2*(A\$7+A\$8))+6	VRAI		
10					
11			ET	VRAI	FAUX
12			VRAI	VRAI	FAUX
13			FAUX	FAUX	FAUX
14					
15					
16		=ET(A\$7>A\$8;A\$8>A\$9)	FAUX		



	A	B	C	D	E
1	Le OU logique :				
2	Deux ou plusieurs critères :				
3	OU(Critère1;Critère2)				
4	OU(Critère1;Critère2; ... ; CritèreN)				
5					
6	Données	Formules	Résultats		
7	17	=A\$7>A\$8	VRAI		
8	9	=A\$8>A\$9	FAUX		
9	58	=A\$9=(2*(A\$7+A\$8))+6	VRAI		
10					
11			OU	VRAI	FAUX
12			VRAI	VRAI	VRAI
13			FAUX	VRAI	FAUX
14					
15					
16		=OU(A\$7>A\$8;A\$8>A\$9)	VRAI		



	A	B	C	D
1	Le NON logique :			
2	Un seul critère :			
3	NON(Critère)			
4				
5				
6	Données	Formules	Résultats	
7	17	=NON(\$A\$7>\$A\$8)	FAUX	
8	9	=NON(\$A\$8>\$A\$9)	VRAI	
9	58	=NON(\$A\$9=(2*(\$A\$7+\$A\$8))+6)	FAUX	
10				
11		↓		NON
12			VRAI	FAUX
13			FAUX	VRAI
14				
15		=\$A\$7<=\$A\$8	FAUX	
16		=\$A\$8<=\$A\$9	VRAI	
17		=\$A\$9<>(2*(\$A\$7+\$A\$8))+6	FAUX	



On peut combiner les conditions avec l'opérateur logique **ET** :

ÉCRITURE ALGORITHMIQUE	ÉCRITURE EXCEL
SI Condition1 et Condition2 ALORS ActionOui FINSI	=SI(ET(Condition1; Condition2) ; ActionOui;"")
SI Condition1 et Condition2 ALORS ActionOui SINON ActionNon FINSI	=SI(ET(Condition1; Condition2) ; ActionOui;ActionNon)
SI Condition1 et Condition2 et Condition3 et ... ALORS ActionOui SINON ActionNon FINSI	=SI(ET(Condition1; Condition2 ; Condition3 ; ...) ; ActionOui;ActionNon)
SI Condition1 et Condition2 et Condition3 et ... ALORS SI ... SINON SI ... FINSI	=SI(ET(Condition1; Condition2 ; Condition3 ; ...) ; SI (...); SI (...))



On peut combiner les conditions avec l'opérateur logique **OU** :

ÉCRITURE ALGORITHMIQUE	ÉCRITURE EXCEL
SI Condition1 Ou Condition2 ALORS ActionOui FINSI	=SI(OU(Condition1; Condition2) ; ActionOui;"")
SI Condition1 Ou Condition2 ALORS ActionOui SINON ActionNon FINSI	=SI(OU(Condition1; Condition2) ; ActionOui;ActionNon)
SI Condition1 Ou Condition2 Ou Conditio3 Ou ... ALORS ActionOui SINON ActionNon FINSI	=SI(OU(Condition1; Condition2 ; Condition3 ; ...) ; ActionOui;ActionNon)
SI Condition1 Ou Condition2 Ou Conditio3 Ou ... ALORS SI ... SINON SI ... FINSI	=SI(OU(Condition1; Condition2 ; Condition3 ; ...) ; SI (...); SI (...))



Les fonctions logiques **ET** / **OU** peuvent être combinées :

Les fonctions logiques ET() et OU() :

ÉCRITURE ALGORITHMIQUE	ÉCRITURE EXCEL
(critère1 ET critère2 ET ...)	ET(critère1;critère2 ; ...)
(critère1 OU critère2 OU ...)	OU(critère1;critère2 ; ...)
(critère11 ET critère12 ET ...) OU (critère21 ET critère22 ET ...)	OU(ET(critère11;critère12 ; ...); ET(critère21;critère22 ; ...))
(critère11 OU critère12 OU ...) ET (critère21 OU critère22 OU ...)	ET(OU(critère11;critère12 ; ...); OU(critère21;critère22 ; ...))



La fonction logique **NON()** permet d'inverser la valeur logique de l'expression logique exprimée (c'est à dire de la condition composée de plusieurs critères).

Opérateurs Logiques / Conditions ou Expressions logiques	
Opérateur de comparaison →>	→> Sa négation (son contraire !)
< <= = >= > <>	>= > <> < <= =
Expression logique →>	→> Sa négation
ET(Critère1;Critère2) ET(Crit1;Crit2; ...; Critn) OU(Crit1;Crit2) OU(Crit1;Crit2; ...; Critn)	OU(NON(Critère1);NON(Critère2)) OU(NON(Crit1);NON(Crit2);...;NON(Critn)) ET(NON(Crit1);NON(Crit2)) ET(NON(Crit1);NON(Crit2);...;NON(Critn))



Exemple:

	A	B	C
1	Numéro	Nom	Age
2	1	Clémence	35
3	2	Adam	22
4			

Traitement à réaliser :

Si le numéro est inférieur à 10 et si l'âge est compris entre 20 et 40

Alors Afficher le message "Nouveau"

Si non Afficher le message "Ancien"

FinSi

Formules XL correspondantes :

Première possibilité :

=SI(**ET(A2<10;C2>=20;C2<=40)** ; "Nouveau" ; "Ancien")

Deuxième possibilité : !!!

=SI(**OU(A2>=10;C2<20;C2>40)** ; "Ancien" ; "Nouveau")



Formules XL correspondantes :

Première possibilité :

=SI(**ET(A2<10;C2>=20;C2<=40)** ; "Nouveau" ; "Ancien")

Deuxième possibilité : !!!

=SI(**OU(A2>=10;C2<20;C2>40)** ; "Ancien" ; "Nouveau")

En appliquant la règle d'inversion suivante :

NON (ET(A2<10;C2>=20;C2<=40))

est équivalent à

OU (NON(A2<10) ; NON(C2>=20) ; NON(C2<=40))

est équivalent à

OU (A2>=10 ; C2<20 ; C2>40)



Inversion de condition :

ET(A2<10;C2>=20;C2<=40) →→→→ Sa négation, son contraire est

NON(ET(A2<10;C2>=20;C2<=40)) =

OU(NON(A2<10);NON(C2>=20);NON(C2<=40)) =

OU(A2>=10;C2<20;C2>40)



Exemple:

	A	B	C
1	Numéro	Nom	Age
2	1	Clémence	35
3	2	Adam	22
4			

Traitement à réaliser :

Si le numéro est supérieur ou égal à 10 ou si l'âge est n'est pas compris entre 20 et 40

Alors Afficher le message "Ancien"

Si non Afficher le message "Nouveau"

FinSi

Formules XL correspondantes :

Première possibilité :

=SI(**OU(A2>=10;C2<20;C2>40)** ; "Ancien" ; "Nouveau")

Deuxième possibilité : !!!

=SI(**NON(OU(A2>=10;C2<20;C2>40))** ; "Nouveau" ; "Ancien")

=SI(**ET(A2<10;C2>=20;C2<=40)** ; "Nouveau" ; "Ancien")



Inversion de condition :

OU(A2>=10;C2<20;C2>40) →→→→ Sa négation, son contraire est

NON(OU(A2>=10;C2<20;C2>40)) =

ET(NON(A2>=10);NON(C2<20);NON(C2>40)) =

ET(A2<10;C2>=20;C2<=40)



2. Liste de quelques opérateurs et fonctions XL très utiles

2.1.

Type	Liste des fonctions en bibliothèque	
	Opérateurs	Fonction(s) possibles cataloguées
Texte	Concaténation &	GAUCHE(), DROITE(), STXT() MINUSCULE(), TEXTE(), REPT(), CONCATENER()...
Date	Addition (nombre de jours) + , Soustraction (nombre de jours) -	DATE(), JOUR(), MOIS(), ANNEE(), AUJOURDHUI()...
Heure	+, -	HEURE(), TEMPS()...
Numérique	Addition + , Soustraction - , Division / , Multiplication * , Puissance ^ , Pourcentage % Attention à la priorité entre les opérateurs	MIN(), MAX(), SOMME(), MOYENNE(), NBVAL(), NB(), ALEA(), ENT(),ARRONDI(), ARRONDI.INF(), ARRONDI.SUP()... SOMME.SI(), NB.SI(), MOYENNE.SI()



2.1.

Type	Liste des fonctions en bibliothèque	
	Opérateurs	Fonction(s) possibles cataloguées
Logique, Bouléen	ET, OU, NON	Toutes, SI(), ESTVIDE(), ESTNUM(), ESTTEXTE(), ESTERREUR(), SIERREUR(°...
Financière Numérique		AMORLIN(), TIR(), VPM()...
Comparaison	Inférieur <, Inférieur ou égal <=, Egal =, Supérieur >, Supérieur ou égal >=, Différent <>	Toutes

Les fonctions XL peuvent être classées de plusieurs manières :

- Classement selon le type du résultat (Texte, Numérique, Date, Logique ...)
- Classement selon les catégories de calculs (sans condition ; sans critère de sélection) ou (avec conditions ; avec critères de sélection)
- Classement selon le nombre d'arguments (de paramètres)
- Classement selon la fonctionnalité (exemple recherche d'information dans un tableau de données) ...



2.2. Attention à la priorité entre les opérateurs

Addition +, Soustraction -, Division /, Multiplication *, Puissance ^, Pourcentage %

$$\begin{array}{lcl} \mathbf{A \times B + C} & = & \mathbf{??? (A \times B) + (A \times C)} \\ \mathbf{A \times B + C} & = & \mathbf{??? (A \times B) + C} \end{array}$$

$$\mathbf{A \times (B + C) = (A \times B) + (A \times C)}$$

$$\mathbf{A \times (B + C / (D + E)) = ???}$$

$$\mathbf{A \times B + C / D + E = ???}$$

$$\mathbf{A \times B + C / (D + E) = ???}$$

Les parenthèses permettent de lever toute ambiguïté ;

la priorité entre les opérateurs aussi :

Parenthèses : commencer par le couple de parenthèses () le plus interne

Priorité : Arithmétique ^ Puissance ;
 * / Multiplication Division ;
 + - Addition Soustraction

Priorité : Logique, Bouléen NON ; ET ; OU



2.3. Notion d'arguments de fonctions

Exemples de fonctions XL du type *Texte* :

GAUCHE(□;□), **DROITE**(□;□), **STXT**(□; □; □),
MINUSCULE(□),**MAJUSCULE**(□)...

Exemples de fonctions XL du type *Date* :

AUJOURDHUI() est un exemple de fonction XL **sans argument** (rien entre les parenthèses, on demande un résultat sans donner de paramètre)

JOUR(unedate), **MOIS**(unedate), **ANNEE**(unedate) sont trois exemples de fonctions XL à **une variable de type date**. Le résultat demandé est fonction d'un paramètre, que l'on donne, qui doit être une date.

DATE(uneannée ; unmois ; unjour), est un exemple de fonction XL à **trois variables de types numériques**. Le résultat demandé est fonction des paramètres, que l'on donne, qui doivent être tous des numériques.



Notion d'argument(s) d'une fonction

Exemples de *fonctions XL* du type *Numérique* :

3 catégories de fonctions :

- Sans argument (avec des parenthèses vides)

Exemples : PI() ALEA()

- Avec arguments (un nombre fixe d'arguments)

Exemples : RACINE(un seul argument)

- Avec arguments (un nombre variable d'arguments)

Exemples : SOMME(un ou plusieurs arguments séparés par ; ou :)

Exemples :

=SOMME(1;2;A4;A7)

=SOMME(SOMME(1;2);A4;A7)

=SOMME(PI();RACINE(MOYENNE(A2:A22));MAX(B7:B77))



Exemples d'arguments de fonction :

=SOMME(1;2;A4;A7)

=SOMME(SOMME(1;2);A4;A7)

=SOMME(SOMME(1;2);SOMME(A4;A7))

=SOMME(1;2)+SOMME(A4;A7)

=SOMME(1;RACINE(4);A4;A7)

=MOYENNE(1;RACINE(4);A4;((RACINE((D6*D7)+4))/D4)+15)

Un argument est donc de plusieurs catégories :

Une valeur

1, 7, etc...

Une cellule

B7, A\$19

Un ensemble de cellules

A1 :B7

Une fonction

PI(), SOMME(A1 :B7 ;C1 :C8)

Une combinaison de ces catégories



2.4. La fonction RECHERCHEV

Elle permet de retrouver dans une **Base de Données (Une feuille XL ou un table) référencée** une ligne afin de compléter l'information. La ligne en question, si elle existe, est reconnue au moyen de la **CléEtr**.

Dans Excel, la fonction RECHERCHEV s'écrit comme suit :

=RECHERCHEV(CléEtr;BDdeRéférence-Feuille-Fichier;NumCol-Résultat;FAUX)

Remarque : Si la clé n'existe pas dans la base de données, Excel renvoi un message d'erreur du type #N/A. Si l'on veut **afficher un message d'erreur personnalisé** pour signaler autrement l'anomalie (l'erreur), il faut utiliser la fonction :

=SIERREUR(FormuleXL;"Message d'erreur à afficher")

=SIERREUR(RECHERCHEV(CléEtr;BDdeRéférence-Feuille-Fichier;NumCol-Résultat;FAUX);"Message d'erreur à afficher")



3. La gestion de l'affichage

Les tableurs proposent plusieurs formats de données :

- des formats prédéfinis (Nombre, Date, Heure, Texte, ...)
- des formats personnalisés (définis par l'utilisateur)
- mise en forme conditionnelle

Pour les données du type numérique, il faut savoir que la donnée affichée à l'écran ne correspond pas forcément à la valeur mémoire réellement prise en compte dans les calculs (attention au problème d'arrondi).

Exemples de formats personnalisés :

Format	Signification	Exemples
0	Numérique sans décimale	1000 50
0,00	Numérique avec 2 décimales	1000,00 50,00
# ##0	Numérique sans décimale avec séparateur de millier	1 000 50
# ##0,0	Numérique avec 1 décimale avec séparateur de millier	1 000,0 50,0
0" Euro(s)"	Afficher le texte entre guillemets après la valeur numérique	1000 Euro(s)
jjj jj mmm aaaa	Afficher la date en toutes lettres (les 3 premières lettres)	jeu 16 oct 2009
jjjj jj mmmm aaaa	Afficher la date en toutes les lettres (toutes lettres)	jeudi 21 octobre 2009



4. Les graphiques

Les graphiques permettent de donner une représentation visuelle très utile pour l'aide à la décision.

En effet, il est fréquent de synthétiser les données sous formes de dessins pour se donner une idée globale des ordres de grandeur des phénomènes étudiés.



	A	B	C	D	E	F	G	H	I	J
1	Taille, en cm, d'un Bébé pendant la première année									
2	Mois	Taille minimale	Taille maximale	Taille du bébé						
3	0	45	55	50						
4	1	48	58	53						
5	2	50	61	55						
6	3	52	64	57						
7	4	54	67	58						
8	5	56	70	61						
9	6	58	72	63						
10	7	60	74	65						
11	8	62	76	66						
12	9	64	78	68						
13	10	65	80	69						
14	11	66	81	70						
15	12	67	83	72						

Courbes représentant la taille d'un BB de 0 à 12 mois

Taille d'un Bébé ... !!!

Graphique INSIGNIFIANT :
la représentation "Secteur 3D" n'est pas la bonne !!!

BONNE REPRESENTATION GRAPHIQUE :
la représentation "courbes" est la bonne !!!



5. Excel et les Bases de Données :

EXCEL offre plusieurs catégories de feuilles :

Feuilles de Calculs (feuilles classiques)

Feuilles de **Commandes, macros** ou modules

→ Feuilles de Calculs pour

Faire des **calculs**

Arithmétiques, Logiques, Financiers, **Statistiques**, etc...

Manipuler des données Classer/trier, Filtrer/Sélectionner,

Insérer, Supprimer, Modifier,

Analyser, etc...

Faire des sous totaux, des rapports de tableaux croisés dynamiques

Gérer un ensemble de **données** (feuilles, tableaux, tables)

Cet ensemble peut être appelé "**une Base de Données**" simplifiée

→ Feuilles de Commandes pour

Faire des **programmes d'applications** :

Gestion de stock, gestion de portefeuille, paye etc...

plusieurs langages de programmation : macro-commandes XL ;

Visual Basic ; autre




Qu'est ce une base de données ?

Une base de données (**BD**) est un ensemble de données, par exemple :

- * Adresses de clients dans un carnet d'adresses
- * Informations sur les ventes dans un livre comptable
- * Formulaire d'informations relatives aux employés dans un dossier

On peut parler d'un ou plusieurs tableaux que l'on appellera **tables de données**.





Les Bases de Données (Feuilles, Tables) permettent de stocker les données relatives à un sujet spécifique.

Une Feuille/Table a la structure suivante :

Table des clients

Numéro	Nom	Prénom	Date de naissance	Chiffre d'Affaires	Téléphone	Email
1rb	RAHMA	Clémence	16/10/1996	19 960	0145750679	rb@world.com
2ab	RAHMA	Adam	19/06/2001	19 000	0606070712	ab@free.tn
3ib	ETINCELLE	Inès	22/11/1969	69 000	0617060707	ib@labas.fr
...	...					

Une **table** correspond à un **tableau de données à deux dimensions** ou **matrice** (avec des lignes et des colonnes). Dans **Excel**, c'est une **feuille de calcul**.



Table1 / Relation1

Table2 / Relation2

Table3 / Relation3

Clé primaire : colonne(s) où les valeurs en double ne sont pas admises

Clé étrangère : colonne(s) dont les valeurs font référence à une clé primaire



Excel offre 3 catégories de fonctions pour manipuler les données dans les tables :

Cat1. **Sans** traitement conditionnel : *aucune sélection à faire sur les données*

Toutes les fonctions classiques d'Excel telles que :

SOMME, NB, MOYENNE, MAX, MIN, ...

GAUCHE, DROITE, STXT, TEXTE, SUBSTITUTE, ...

DATE, MOIS, JOUR, ANNEE, ...,

AMORLIN,



Excel offre 3 catégories de fonctions pour manipuler les données dans les tables :

Cat2. **Avec** traitement(s) conditionnel(s) ; **Sans** techniques matricielles (sans les { }):
On sélectionne les données sur lesquelles on effectue les calculs
→ MONO-CRITERE

Seulement **trois** fonctions

NB.**SI**(ZoneATester;LeCritère)

SOMME.**SI**(ZoneAtester;LeCritère;ZoneDeCalcul) /

SOMME.**SI**(ZoneAtester;LeCritère)

MOYENNE.**SI**(ZoneAtester;LeCritère;ZoneDeCalcul) /

MOYENNE.**SI**(ZoneAtester;LeCritère)



Excel offre 3 catégories de fonctions pour manipuler les données dans les tables :

Cat3a. Avec traitement(s) conditionnel(s) ;

On sélectionne les données sur lesquelles on effectue les calculs

→ MONO-CRITERE ET MULTI-CRITERES

**NB.SI.ENS()
SOMME.SI.ENS()
MOYENNE.SI.ENS()**



Cat3b. Avec traitement(s) conditionnel(s) ; Avec techniques matricielles (avec les { }) :
On sélectionne les données sur lesquelles on effectue les calculs


→ MONO-CRITERE ET MULTI-CRITERES

Plusieurs fonctions

```
{=NB(SI( Condition ;1;""))}
{=SOMME(SI( Condition ;ZoneDeCalcul;""))}
{=MIN(SI(Condition;ZoneDeCalcul;""))}
{=MAX(SI(Condition;ZoneDeCalcul;""))}
{=MOYENNE(SI(Condition;ZoneDeCalcul;""))}
{=VARP(SI(Condition;ZoneDeCalcul;""))} etc....
```



6. Les fonctions **NB.SI()**, **SOMME.SI()** et **MOYENNE.SI()** : (les fonctions « deux en un »)



Exemples : **Comment écrire LeCritère ???**

" > 20 "

" >= 77 "

" <> 55 "

" < 10 "

" <= 77 "

" = 55 "

" = 10 "

" 10 "

Attention le égal est implicite

" = Adam "

" Adam "

Attention pas de guillemets pour les Textes

" = Clémence "

" Clémence "

Attention pas de guillemets pour les Textes

A7

Renvoyer à la cellule A7 par exemple

Nombre de notes supérieures à 10 :

NB.SI(Notes ; " > 10 ")

Nombre de notes inférieures à 10 :

NB.SI(Notes ; " < 10 ")

Nombre de notes égales à 10 :


NB.SI(Notes ; " = 10 ")

Nombre de notes égales à 10 :

NB.SI(Notes ; " 10 ")



La fonction SOMME.SI(ZoneATester; **LeCritère** ; ZoneDeCalcul)



Exemples : **Comment écrire LeCritère ???**


Total du Chiffre d'affaires de Clémence : **SOMME.SI(Nom ; " = Clémence " ; CA)**

Cumul des Chiffres d'affaires > 10000 : **SOMME.SI(CA ; " > 10000 " ; CA)**

Cumul des Chiffres d'affaires <> 10000 : **SOMME.SI(CA ; " <> 10000 ")**



La fonction MOYENNE.SI(ZoneATester; **LeCritère** ;ZoneDeCalcul)



Exemples : **Comment écrire LeCritère ???**

Moyenne du Chiffre d'affaires de Clément : MOYENNE.SI(Nom ; " = Clément " ; CA)

Moyenne des Chiffres d'affaires > 10000 : MOYENNE.SI(CA ; " > 10000 " ; CA)



Les fonctions **NB.SI**, **SOMME.SI** et **MOYENNE.SI** ne peuvent porter que sur **UN SEUL critère (mono-critère)**.

NB.SI(ZoneATester;LeCritère)

SOMME.SI(ZoneAtester;LeCritère;ZoneDeCalcul) /
 SOMME.SI(ZoneAtester;LeCritère)

MOYENNE.SI(ZoneAtester;LeCritère;ZoneDeCalcul) /
 MOYENNE.SI(ZoneAtester;LeCritère)

Des critères simples sans utiliser d'autres fonctions XL telles que :
 GAUCHE, DROITE, STXT, TEXTE et DATE.



Exemples : Comment écrire LeCritère ???

" A* "

commence par

" *A "

se termine par

" *A* "

contient

Le caractère * joue le rôle du JOCKER !

Exemples :

commence par de(s) lettre(s)

se termine par de(s) lettre(s)

contient de(s) lettre(s)



Le caractère * joue le rôle du JOCKER !

Exemples :

Chiffre d'affaires des clients dont le **nom commence par la lettre C :**

SOMME.SI(Nom ; " C* " ; CA)

Chiffre d'affaires des clients dont le **nom se termine par la lettre C :**

SOMME.SI(Nom ; " *C " ; CA)

Chiffre d'affaires des clients dont le **nom contient la lettre C :**

SOMME.SI(Nom ; " *C* " ; CA)



Les fonctions **NB.SI**, **SOMME.SI** et **MOYENNE.SI** ne peuvent porter que sur **UN SEUL** critère (mono-critère).

Le critère doit être simple.

Il ne peut contenir qu'une seule comparaison : **mono-critère**

Il est possible d'utiliser le **caractère *** pour répondre à des questions du style :
commence par, se termine par, contient.

Il n'est pas possible d'utiliser, dans la formule, les fonctions excel telles que :

~~GAUCHE, DROITE, STXT, TEXTE et DATE.~~

~~NB.SI (GAUCHE(NOM ;1) ; " A ")~~



7. Les fonctions **NB.SI.ENS()**, **SOMME.SI.ENS()** et **MOYENNE.SI.ENS ()** :

Les fonctions **NB.SI.ENS**, **SOMME.SI.ENS** et **MOYENNE.SI.ENS** peuvent porter que sur **PLUSIEURS** critères.

NB.SI.ENS(Zone1ATester; LeCritère1 ; Zone2ATester; LeCritère2...Etc)



8. Les techniques matricielles :

Les calculs conditionnels ; les fonctions « deux en un »

FonctionXL (**SI** (**Condition-Test** ; **AlorsActionduOui** ; **SinonActionduNon**))

SOMME(**SI**(... ; ... ; ...))

NB(**SI**(... ; ... ; ...))

MOYENNE(**SI**(... ; ... ; ...))

MAX(**SI**(... ; ... ; ...))

MIN(**SI**(... ; ... ; ...))

VARP(**SI**(... ; ... ; ...)) ... etc



Les **techniques matricielles** permettent de réaliser des **traitements conditionnels** sur un ensemble de données **-une matrice ou un tableau de données ou une table de données-**. Les calculs possibles utilisent les fonctions telles que

SOMME, NB, MOYENNE, MAX, MIN, VARP etc.

Ces fonctions sont combinées avec la fonction **SI**.

Ces fonctions permettent d'exprimer des conditions

- (1) **simples mono-critère** et
- (2) des conditions **complexes multi-critères**.

Ces techniques permettent de **faire des analyses de données**, d'**établir des tableaux de bord** et **des graphiques** : Une bonne exploitation des données pour aider à la prise de décisions et donc manipuler les bases de données.

C'est un complément idéal pour la commande ("le menu Données" d'EXCEL) sur les « sous totaux » et les « **tableaux croisés dynamiques** ».



Pour illustrer ce type de calcul, nous allons analyser les données clients ci-dessous.

Table des clients

Numéro	Nom	Prénom	Date de naissance	Chiffre d'Affaires	Téléphone	Email
1rb	RAHMA	Clémence	16/10/1996	19 960	0145750679	rb@world.com
2ab	RAHMA	Adam	19/06/2001	19 000	0606070712	ab@free.tn
3ib	ETINCELLE	Inès	22/11/1969	69 000	0617060707	ib@labas.fr
...	...					

Sous Excel cette table sera présentée dans une feuille d'un classeur comme suit :

	A	B	C	D	E	F	G
1	Numéro	Nom	Prénom	Date de naissance	Chiffre d'Affaires	Téléphone	Email
2	1rb	RAHMA	Clémence	16/10/1996	19 960	0145750679	rb@world.com
3	2ab	RAHMA	Adam	19/06/2001	19 000	0606070712	ab@free.tn
4	3ib	ETINCELLE	Inès	22/11/1969	69 000	0617060707	ib@labas.fr
5	...						
6					



Exemples de requêtes (questions formulées sur la table de la base de données) :

Requêtes NON sélectives

RA1 : Nombre de client(s) ?

RA2 : Nombre d'affaire(s) ?

RA3 : Total des chiffres d'affaires ?

RA4 : Moyenne des chiffres d'affaires ?

RA5 : Le plus grand chiffre d'affaires ?

RA6 : Le plus petit chiffre d'affaires ?



Exemples de requêtes (questions formulées sur la table de la base de données) :

Requêtes sélectives MONO-CRITERE

RB1 : Nombre de client(s) de nom RAHMA?

RB2 : Nombre d'affaire(s) d'un montant > 20000 ?

RB3 : Total des chif. d'aff. de la cliente dont le prénom est Inès ?

RB4 : Moyenne des chif. d'aff. des clients de nom RAHMA ?

RB5 : Le plus grand chif. d'aff. des clients de nom RAHMA ?

RB6 : Le plus petit chif. d'aff. des clients de nom RAHMA ?

RB7 : Total des chif. d'aff. des clients dont le prénom commence par la lettre A ?

RB8 : Total des chif. d'aff. des clients dont le prénom se termine par la lettre A ?

RB9 : Total des chif. d'aff. des clients dont le prénom contient la lettre A ?

RB10 : Nombre de client(es) né(es) avant l'an 2000 ?



Exemples de requêtes (questions formulées sur la table de la base de données) :

Requêtes sélectives MULTI-CRITERES

RC1 : Nombre de client(s) de nom RAHMA et de prénom ADAM ?

RC2 : Nombre d'affaire(s) d'un montant compris entre 20000 et 30000 ?

RC3 : Total des chif. d'aff. > 20000 de la cliente dont le prénom est Inès ?

RC4 : Moyenne des chif. d'aff. > 20000 des clients de nom RAHMA ?

RC5 : Le plus grand chif. d'aff. > 20000 des clients de nom RAHMA ?

RC6 : Le plus petit chif. d'aff. > 20000 des clients de nom RAHMA ?

RC7 : Total des chif. d'aff. des clients dont le nom ou le prénom contient la lettre A ?

RC8 : Nombre de clients dont la dernière lettre du nom est la même que la première lettre du prénom ?



Résumé :

plusieurs catégories de fonctions sélectives :

La première catégorie ne permet que des sélections **mono-critère** !

=NB.SI(ZoneAtester;LeCritère)

=SOMME.SI(ZoneAtester;LeCritère;ZoneDeCalcul)

=MOYENNE.SI(ZoneAtester;LeCritère;ZoneDeCalcul)

Ces fonctions ne permettent de traiter que des **conditions MONO-CRITERE**



La deuxième catégorie permet des sélections **mono-critère** et **multi-critères** !

```
=NB.SI.ENS(Zone1Atester;LeCritère1; Zone2Atester;LeCritère2; Zone3Atester;LeCritère3...)
```

```
=SOMME.SI.ENS(...)
```

```
=MOYENNE.SI.ENS(...)
```



La troisième catégorie permet des sélections **mono-critère** et **multi-critères** !

`{=SOMME(SI(Condition ;ActionOui ;ActionNon))}`

`{=NB(SI(Condition ;ActionOui ;ActionNon))}`

`{=MOYENNE(SI(Condition ;ActionOui ;ActionNon))}`

`{=MAX(SI(Condition ;ActionOui ;ActionNon))}`

`{=MIN(SI(Condition ;ActionOui ;ActionNon))}`

`{=VARP(SI(Condition ;ActionOui ;ActionNon))}`

L'action N°1 « ActionOui » est soit le nom de la zone de calcul (pour les fonctions autres que NB) soit la valeur 1 (pour la fonction NB)

L'action N°2 « ActionNon » peut être vide (représentée sous EXCEL par un -double guillemets « ») :



{=SOMME(SI(Condition ;ActionOui ; « »))}

{=NB(SI(Condition ;ActionOui ; « »))}

{=NB(SI(Condition ;1 ; « »))}

{=MOYENNE(SI(Condition ;ActionOui ; « »))}

{=MAX(SI(Condition ;ActionOui ; « »))}

{=MIN(SI(Condition ;ActionOui ; « »))}

{=VARP(SI(Condition ;ActionOui ; « »))}

Ces fonctions permettent de traiter
 des conditions **MONO-CRITERE** et des conditions **MULTI-CRITERES**



Dans le cas où la condition porte sur plusieurs critères deux opérateurs logiques peuvent être utilisés : le ET (remplacé ici par le * !!!) et le OU (remplacé ici par le + !!!) comme suit :

ÉCRITURE ALGORITHMIQUE	ÉCRITURE EXCEL
(critère1 ET critère2 ET ...)	(critère1)*(critère2)* ...
(critère1 OU critère2 OU ...)	(critère1)+(critère2)+ ...
(critère11 ET critère12 ET ...) OU (critère21 ET critère22 ET ...)	((critère11)*(critère12)..) + ((critère21)*(critère22)..)
(critère11 OU critère12 OU ...) ET (critère21 OU critère22 OU ...)	((critère11)+(critère12)..) * ((critère21)+(critère22)..)
Négation(critère)	NON(Critère)
Négation(plusieurs critères)	NON(Critère1 ; Critère2 ; ...)

Les conditions peuvent porter sur un ou plusieurs critères (**mono-critère et/ou multi-critères**)

Il n'est pas possible d'utiliser le caractère * pour répondre à des questions du style :
 commence par, se termine par, contient.

Il est possible d'utiliser, dans la formulation de la condition, les fonctions excel du type :
 gauche, droite, texte, date, ...



Les fonctions logiques ET et OU sont remplacées par * et +

ET(Critère1;Critère2...) est remplacé par (Critère1)*(Critère2)*

OU(Critère1;Critère2...) est remplacé par (Critère1)+(Critère2)+

Exemple :

{=NB(SI((critère1)+((critère2)*(critère3)) ;1;""))}

Comment calculer une moyenne :

Est-il possible de faire le calcul ? (Y a-t-il des réponses dans ma base de données ?)

Si oui

Alors faire le calcul de la moyenne

Sinon afficher Impossible ou pas de réponse

Fin Si

Traduction sous EXCEL :

=SI(NB.SI(...;...) > 0 ; SOMME.SI(...;...;...)/NB.SI(...;...) ; "Pas de réponse")

=SI(NB.SI(...;...) = 0 ; "Pas de réponse" ; SOMME.SI(...;...;...)/NB.SI(...;...))

{=SI(NB(SI(...;1;"")) > 0 ; MOYENNE(SI(...;...;"")) ; "Pas de réponse")}

{=SI(NB(SI(...;1;"")) = 0 ; "Pas de réponse" ; MOYENNE(SI(...;...;"")))}



Aujourd'hui, j'arrête de fumer !

Le TABAC t'ABAT !

**FUMER NUIT GRAVEMENT À TA SANTÉ ET
À CELLE DE TON ENTOURAGE !**

M T Dents !

SMILE and the World SMILES with you !



The End : Thanks for your attention

Fin : Merci de votre attention

النهاية : شكرا لحسن إنتباهكم

Fin : Gracias por su atención

Ende : Danke für Ihre Aufmerksamkeit

Конец : Спасибо за ваше внимание

結束 : 感謝您的關注



The End : Thanks for your attention

Fin : Merci de votre attention


النهاية: شكرا لحسن إنتباهكم

Fin : Gracias por su atención

Ende : Danke für Ihre Aufmerksamkeit

Конец : Спасибо за ваше внимание

結束 : 感謝您的關注



M. Faouzi BOUFARES

IUT Villetaneuse, Bureau Q304

boufares@iutv.univ-paris13.fr

Institut Galilée, Bureau A109

boufares@lipn.univ-paris13.fr

Laboratoire d'Informatique de Paris Nord

LIPN – UMR 7030 du CNRS

Université Paris 13. Sorbonne Paris Cité

F-93430 Villetaneuse FRANCE

www.univ-paris13.fr

